
[bookmark: clause_Templates_GlobalLocal][bookmark: _Toc7508627]15.3	Global and local templates
TTCN-3 allows defining global templates and local templates.
Syntactical Structure
template [ restriction ] [ @fuzzy ] [ @deterministic ] [ @abstract ] Type TemplateIdentifier
["(" TemplateFormalParList ")"] [ modifies TemplateRef ] ":=" BaseTemplateBody

NOTE:	The optional restriction part is covered by clause 15.8.
Semantic Description
Global templates shall be defined in the module definitions part. Local templates shall be defined in testcases, functions, altsteps or statement blocks. Both global and local templates shall adhere to the scoping rules specified in clause 5.
Both global and local templates can be parameterized. The actual parameters of a template can include values and templates. The rules for formal and actual parameter lists shall be followed as defined in clause 5.2.
Both global and local templates are initialized at the place of their declaration. This means, all template fields which are not affected by parameterization shall receive a value or matching mechanism. Template fields affected by parameterization are initialized at the time of template use.
If functions are used for the initialization of module parameters, it is strongly advised to adhere to the rules defined in clause 16.1.4. Not following these rules may cause non-deterministic test executions.
At the time of their use (e.g. in communication operations send, receive, call, getcall, etc.), it is allowed to change template fields by in-line modified templates, to pass in values via value parameters as well as to pass in templates via template parameters. The effects of these changes on the values of the template fields do not persist in the template subsequent to the corresponding communication event.
Restrictions
In addition to restrictions in clause 15, the following restrictions apply:
a)	The dot notation such as myTemplateId.fieldId shall not be used to set or retrieve values in templates in communication events. The "->" symbol shall be used for this purpose (see clause 23).
b)	Restrictions on referencing elements of templates or template fields are described in clause 15.6.
c)	There exist a number of restrictions on the functions used in expressions when specifying templates or template fields; these are specified in clause 16.1.4.
d)	After completing initialization, global and local templates that do not contain the @abstract modifier shall be fully initialized.
	NOTE: Initialization of templates without parameters is completed at the place of their declaration. Initialization of parameterized templates is completed at the time of their use.
Examples
	// The template 
	template MyMessageType mw_myTemplate (integer p_myFormalParam):=
	{	
		field1 := p_myFormalParam,
		field2 := pattern "abc*xyz",
		field3 := true
	}

	// could be used as follows 
	pco1.receive(mw_myTemplate(123)); 


[bookmark: clause_Templates_Modifiedl][bookmark: _Toc7508629]15.5	Modified templates
In cases where small changes are needed to specify a new template, it is possible to specify a modified template. A modified template specifies modifications to particular fields of the original template, either directly or indirectly. As well as creating explicitly named modified templates, TTCN‑3 allows the definition of in-line modified templates.
Syntactical Structure
Global or local modified template:
template [restriction] [ @fuzzy ] [ @deterministic ] [ @abstract ] Type TemplateIdentifier
["(" TemplateFormalParList ")"] modifies TemplateRef ":=" BaseTemplateBody

NOTE 1:	The optional restriction part is covered by clause 15.8.
In-line modified template:
modifies BaseTemplateBody ":=" BaseTemplateBody

Semantic Description
The modifies keyword denotes the parent template from which the new modified template shall be derived. This parent template may be either an original template or a modified template or also a matching mechanism.
In case that the BaseTemplateBody on the right hand side of the modified template contains references to variables, these are evaluated before the modification and any uninitialized fields or elements in these variables are treated as unspecified, i.e. as if specified with the not used symbol "‑" for the following modification algorithm.
The modifications occur in a linked fashion, eventually tracing back to the original template:
a)	In case of templates, template fields or list elements of simple types, union and enumerated types, the matching mechanism specified in the modified template is simply replacing its corresponding content in its parent.
b)	For templates, template fields and elements of record and set types, if a record or set field and its corresponding matching mechanism is specified in the modified template, then the specified matching mechanism replaces the one specified in the corresponding field of the parent template. If a record or set field or its corresponding matching mechanism is – implicitly or explicitly by using the not used symbol "‑" - left unspecified in the modified template, then the matching mechanism in the corresponding field of the parent template shall be used. When the field to be modified is nested within a template field which is a structured field itself, no other field of the structured field is changed apart from the explicitly denoted one(s).
c)	For templates, template fields and elements of record of and set of types, the above rules specified for records and sets apply with the following deviations:
if the value list notation is used, only the number of elements listed in the modified template is inherited from the parent (i.e. the list is truncated at the last element of the list notation in the modified template);
when individual values of a modified template or a modified template field of record of or set of type wished to be changed, and only in these cases, the index assignment notation may also be used, where the left hand side of the assignment is the index of the element to be altered.
In case of record of and set of types first apply rule (c) to the complete structure (e.g. truncation) than apply further rules for the remaining individual type structure elements (see example 3).
Formal value or template parameters of modified templates inherit the default value or respectively template of the corresponding parameter of their parent templates only, if this is denoted by the dash (don't change) symbol at the place of the parameters' default value or respectively template.
Modified templates may also be restricted. Template restrictions are specified in clause 15.8.
A modified template may also be declared fuzzy using the @fuzzy modifier.
NOTE 2:	If a fuzzy modified template modifies a non-fuzzy unparameterized template, the inherited fields before modification will be the same for every evaluation of the fuzzy template.
Restrictions
In addition to the general static rules of TTCN‑3 given in clause 5, the following restrictions apply:
a)	A modified template shall not refer to itself, either directly or indirectly, i.e. recursive derivation is not allowed.
b)	If a base template has a formal parameter list, the following rules apply to all modified templates derived from that base template, whether or not they are derived in one or several modification steps:
1)	the derived template shall not omit parameters and change types or names of parameters defined at any of the modification steps between the base template and the actual modified template;
2)	a template parameter restriction of a derived template specified at any of the modification steps between the base template and the actual modified template can be changed to a stricter one (see clause 15.8);
3)	a derived template can have additional (appended) parameters if wished;
4)	if the dash (don't change) symbol is used at the place of a default value or default template, the corresponding parameter of the parent template shall have a valid default value or default template, either assigned directly or inherited. If not, this shall cause an error.
c)	Restrictions on referencing elements of templates or template fields are described in clause 15.6: for modified templates the rules for the left hand side of assignments apply.
d)	Limitations on template restrictions described in clause 15.8 shall apply.
e)	After completing initialization, modified global and local templates that do not contain the @abstract modifier shall be fully initialized.
NOTE: Initialization of modified templates without parameters is completed at the place of their declaration. Initialization of modified parameterized templates is completed at the time of their use.
Examples
EXAMPLE 1:	Modifying record templates (non-embedded case)
	// Modifying records
	type record MyRecordType
	{
		integer field1 optional,
		charstring field2,
		boolean field3
	}
	template MyRecordType m_myRecTemplate1 :=
	{
		field1 := 123,
		field2 := "A string",
		field3 := true
	}
	// then writing 
	template MyRecordType m_myRecTemplate2 modifies m_myRecTemplate1 :=
	{
		field1 := omit,					// field1 is optional but present in m_myTemplate1
		field2 := "A modified string"
										// field3 is unchanged
	}
	// is the same as writing 
	// template MyRecordType m_myRecTemplate2 :=
	// {
	//	  field1 := omit,
	//	  field2 := "A modified string",
	//	  field3 := true
	// }

	template MyRecordType m_myRecTemplate3 modifies m_myRecTemplate1 := {omit, "A modified string"}
	//field3 is implicitly left unchanged;
	//m_myRecTemplate3 has the same content as m_myRecTemplate2

	template MyRecordType m_myRecTemplate4 modifies m_myRecTemplate1 := {omit,"A modified string",-}
	//field3 is explicitly left unchanged;
	//m_myRecTemplate4 has the same content as m_myRecTemplate2 and m_myRecTemplate3

EXAMPLE 2:	Modifying record of templates (non-embedded case)
	type record of integer MyRecordOfType;

	template MyRecordOfType m_myBaseTemplate := { 0, 1, 2, 3, 4, 5, 6, 7, 8, 9 };

	template MyRecordOfType m_myRecOfTemplate1 modifies m_myBaseTemplate := 
                                                                 { -, -, 3, 2, -, -, -, -, -, - };
	// m_myRecOfTemplate1 contains { 0, 1, 3, 2, 4, 5, 6, 7, 8, 9 }

	template MyRecordOfType m_myRecOfTemplate2 modifies m_myBaseTemplate := { -, -, 3, 2 };
	// m_myRecOfTemplate2 replaces m_myBaseTemplate with: { 0, 1, 3, 2 };
	// elements 5 to 10 of m_myBaseTemplate are truncated

	template MyRecordOfType m_myRecOfTemplate3 modifies m_myBaseTemplate := { [2] := 3, [3] := 2 };
	// m_myRecOfTemplate3 has the same content as m_myMod1Template: { 0, 1, 3, 2, 4, 5, 6, 7, 8, 9 }

EXAMPLE 3:	Modifying embedded record and record of templates
	//Modifying a record embedded in a record of
	type record of record {
	  integer a,
	  integer b
	} MyListType

	template MyListType mw_myBaseListTemplate := { ?, { a := 1, b := 2 }, ?, { a := 3, b := 4 } }

	template MyListType mw_myListTemplate1 modifies mw_myBaseListTemplate := { [1] := { a := 42 } }
	//Content of field ″a″ of the second element is modified,
	//the content of mw_myListTemplate1 is: { ?, { a := 42, b := 2 }, ?, { a := 3, b := 4 } }

	template MyListType mw_myListTemplate2 modifies mw_myBaseListTemplate := { -, { a := 42 } ,- }
	//Content of field ″a″ of the second element is modified, and the 
	//record of is truncated after the third element: { ?, { a := 42, b := 2 }, ? }

EXAMPLE 4:	Modified in-line template
	// Given 
	template MyRecordType m_setup :=
	{
		field1 := 75,
		field2 := "abc",
		field3 := true
	}

	// Could be used to define an in-line modified template of Setup
	// pco1.send (modifies m_setup := {field1:= 76});

EXAMPLE 5:	Modified parameterized template
	// Given 
	template MyRecordType m_myTemplate1(integer p_myPar):=
	{
		field1 := p_myPar,
		field2 := "A string",
		field3 := true
	}

	// then a modification could be 
		template MyRecordType m_myTemplate2(integer p_myPar) modifies m_myRecTemplate1 :=
		// field1 is parameterized in m_myTemplate1 and remains also parameterized in m_myTemplate2
	{
		field2 := "A modified string"
	}

EXAMPLE 6:	Default values of modified parameterized templates
	// Given 
	template MyRecordType m_myTemplate11 (integer p_int := 5 ):=
	  // p_int has the default value 5
	{
		field1 := p_int,
		field2 := "A string",
		field3 := true
	}

	// then possible template modifications are 
	template MyRecordType m_myTemplate12(integer p_int) modifies m_myTemplate11 :=
	  // p_int had a default value in m_myTemplate11 but has none in this template
	{
		field2 := "B string"
	}

	template MyRecordType m_myTemplate13(integer p_int := 0) modifies m_myTemplate12 := { }
	  // p_int has the default value 0
	  // no change is made to the template's content, but only to the default value of p_int

	template MyRecordType m_myTemplate14(integer p_int := - ) modifies m_myTemplate13 :=
	  // p_int inherits the default value 0 from its parent m_myTemplate13
	{
	  field2 := "C string"
	}

	template MyRecordType m_myTemplate15(integer p_int := - ) modifies m_myTemplate14 :=
	  // p_int inherits the default value 0 from m_myTemplate13 via m_myTemplate14
	{
	  field2 := "D string"
	}

	template MyRecordType m_myTemplate16(integer p_int) modifies m_myTemplate15 := { }
	  // p_int has no default value; no change in the template's content


	template MyRecordType m_myTemplate17(integer p_int := - ) modifies m_myTemplate16 :=
	  // causes an error as p_int has no default value in the parent template m_myTemplate16
	{
	  field2 := "E string"
	}

EXAMPLE 7:	Modifies with variables
	var template MyRecordType v_modification :=
	{
	  field2 := "G string" // field1/field3 are uninitialized
	}
	var template MyRecordType v_myTemplate := modifies m_myTemplate1(5) := v_modification; 
	// no modification of field1/field3
	// results in { field1 := 5, field2 := "G string", field3 := true }

[bookmark: _Toc7508789]A.1.5	TTCN‑3 terminals
[bookmark: _Toc7508790]A.1.5.0	General
TTCN‑3 terminal symbols and reserved words are listed in tables A.2 and A.3.
[bookmark: annex_BNF_SpecialTerminalSymbols]Table A.2: List of TTCN‑3 special terminal symbols
	Begin/end block symbols
	{      }

	Begin/end list symbols
	(      ) 

	Element specifier symbols
	[      ]

	Range symbol
	..

	Line and block comments
	/*    */      //

	Statement separator symbol
	;

	Arithmetic operator symbols
	+     /       -      *

	Concatenation operator symbol
	&

	Relational operator symbols
	!=    ==   >=   <=   <   >

	Shift operator symbols
	<<   >>

	Rotate operator symbols
	<@  @>

	String enclosure symbols
	"       '   

	Wildcard/matching symbols
	?      *  

	Assignment symbol
	:= 

	Communication operation assignment 
	->

	Bitstring, hexstring and Octetstring values
	B     H    O 

	Float exponent
	E

	List element separator symbol
	,

	Field reference
	.

	Decoded field reference
	=>



The predefined function identifiers defined in table 15 and described in annex C shall also be treated as reserved words.
Table A.3: List of TTCN‑3 terminals which are reserved words
	action
activate
address
alive
all
alt
altstep
and
and4b
any
anytype

bitstring
boolean
break

case
call
catch
char
charstring
check
clear
complement
component
connect
const 
continue
control
create

deactivate
decmatch
default
disconnect
display
do
done

else
encode
enumerated
error
except
exception
execute
extends
extension
external
	fail
false
float
for
friend
from
function

getverdict
getcall
getreply
goto
group

halt
hexstring

if
ifpresent
import
in
inconc
infinity
inout
integer
interleave

kill
killed

label
language
length
log

map
match
message
mixed
mod
modifies
module
modulepar
mtc
	noblock
none
not
not_a_number
not4b
nowait
null

octetstring
of
omit
on
optional
or
or4b
out
override

param
pass
pattern
permutation
port
present
private
procedure
public

raise
read
receive
record

recursive
rem
repeat
reply
return
running
runs
	select
self
send
sender
set
setencode
setverdict
signature
start
stop
subset
superset
system

template
testcase
timeout
timer
to
trigger
true
type

union
universal
unmap

value
valueof
var
variant
verdicttype

while
with

xor
xor4b



The TTCN‑3 terminals listed in table A.3 shall not be used as identifiers in a TTCN‑3 module. These terminals shall be written in all lowercase letters.
Additionally, there are special TTCN-3 terminals consisting of an @-symbol, directly followed by an identifier. These terminals shall also be written in all lowercase letters.
NOTE:	These terminals can be used in combination with the @-symbol, which results in a specific semantics for the annotated language element. They can also be used like any other identifier without any special meaning.
Table A.4: List of TTCN‑3 terminals which are modifiers
	@abstract
@control
@decoded
@default
@deterministic 
	@decoded
@default
@deterministic 
@fuzzy
@index
	@fuzzy
@index 
@lazy
@local
	@lazy
@local 
@nocase



[bookmark: tab_Keywords_InExtensions]Table A.5: List of TTCN‑3 terminals which are reserved words in extension packages
	apply
assert
at

configuration
conjunct
cont

delta
disjunct
duration

finished

	history

implies
inv

mode

notinv
now

onentry
onexit

	par
prev

realtime

seq
setstate
static
stepsize
stream

	
timestamp

until

values

wait




The TTCN‑3 terminals listed in table A.5 are used as keywords inside the TTCN-3 extension packages. Using these terminals in the code is not recommended as it might lead to issues in the future.
These terminals shall be written in all lowercase letters.

[bookmark: _Toc7508798]A.1.6.1.3	Template definitions
[bookmark: TTemplateDef]TemplateDef ::= TemplateKeyword [TemplateRestriction] [FuzzyModifier [DeterministicModifier]]
                    [AbstractModifier] BaseTemplate [DerivedDef] AssignmentChar BaseTemplateBody 
[bookmark: TBaseTemplate]BaseTemplate ::= (Type | Signature) Identifier ["(" TemplateOrValueFormalParList   
                                                    ")"] 
[bookmark: TTemplateKeyword]TemplateKeyword ::= "template" 
[bookmark: TDerivedDef]DerivedDef ::= ModifiesKeyword ExtendedIdentifier 
[bookmark: TModifiesKeyword]ModifiesKeyword ::= "modifies" 
[bookmark: TTemplateOrValueFormalParList]TemplateOrValueFormalParList ::= TemplateOrValueFormalPar {"," TemplateOrValueFormalPar} 
[bookmark: TTemplateOrValueFormalPar]TemplateOrValueFormalPar ::= FormalValuePar | FormalTemplatePar 
/* STATIC SEMANTICS - FormalValuePar shall resolve to an in parameter */ 
TemplateBody ::= DerivedTemplateBody | BaseTemplateBody
[bookmark: TTemplateBody]BaseTemplateBody ::= (SimpleSpec | 
                      FieldSpecList | 
                      ArrayValueOrAttrib 
                     ) [ExtraMatchingAttributes] 

/* STATIC SEMANTICS - Within BaseTeplateBody the ArrayValueOrAttrib can be used for array, record, record of and set of types. */ 
[bookmark: TSimpleSpec]SimpleSpec ::= (SingleExpression ["&" SimpleTemplateSpec]) | SimpleTemplateSpec 
[bookmark: TSimpleTemplateSpec]SimpleTemplateSpec ::= SingleTemplateExpression ["&" SimpleSpec] 
[bookmark: TSingleTemplateExpression]SingleTemplateExpression ::= MatchingSymbol | 
                               ({TemplateRefWithParList [ExtendedFieldReference]) |
                                ExtendedIdentifier EnumTemplateExtension
/** STATIC Semantics: ExtendedIdentifier shall refer to an enumerated value with associated value */
[bookmark: TEnumTemplateExtension]EnumTemplateExtension ::= "(" (BaseTemplateBody | Range) {"," (BaseTemplateBody | Range) } ")"
/** STATIC Semantics: each TemplateBody shall be an integer template template and the limits of each Range  an integer value*/
[bookmark: TFieldSpecList]FieldSpecList ::= "{" FieldSpec {"," FieldSpec} "}" 
[bookmark: TFieldSpec]FieldSpec ::= FieldReference AssignmentChar (TemplateBody | Minus) 
[bookmark: TFieldReference]FieldReference ::= StructFieldRef | 
                       ArrayOrBitRef | 
                       ParRef 
[bookmark: TStructFieldRef]StructFieldRef ::= Identifier | 
                       PredefinedType | 
                       TypeReference 

/* STATIC SEMANTICS - PredefinedType and TypeReference shall be used for anytype value notation only. PredefinedType shall not be AnyTypeKeyword.*/ 
[bookmark: TParRef]ParRef ::= Identifier 

/* STATIC SEMANTICS - Identifier in ParRef shall be a formal parameter identifier from the associated signature definition */ 
[bookmark: TArrayOrBitRef]ArrayOrBitRef ::= "[" FieldOrBitNumber "]" 

/* STATIC SEMANTICS - ArrayRef shall be optionally used for array types and TTCN-3 record of and set of. The same notation can be used for a Bit reference inside an TTCN-3 charstring, universal charstring, bitstring, octetstring and hexstring type */ 
[bookmark: TFieldOrBitNumber]FieldOrBitNumber ::= SingleExpression 

/* STATIC SEMANTICS - SingleExpression will resolve to a value of integer type */ 
[bookmark: TArrayValueOrAttrib]ArrayValueOrAttrib ::= "{" [ArrayElementSpecList] "}" 
[bookmark: TArrayElementSpecList]ArrayElementSpecList ::= ArrayElementSpec {"," ArrayElementSpec} 
[bookmark: TArrayElementSpec]ArrayElementSpec ::= Minus | 
                          PermutationMatch | 
                          TemplateBody 
[bookmark: TMatchingSymbol]MatchingSymbol ::= Complement | 
                        (AnyValue [WildcardLengthMatch]) | 
                        (AnyOrOmit [WildcardLengthMatch]) | 
                        ListOfTemplates | 
                        Range | 
                        BitStringMatch | 
                        HexStringMatch | 
                        OctetStringMatch | 
                        CharStringMatch | 
                        SubsetMatch | 
                        SupersetMatch |
                        DecodedContentMatch
[bookmark: TDecodedContentMatch]DecodedContentMatch ::= DecodedMatchKeyword ["(" [Expression] ")"] TemplateInstance
[bookmark: TDecodedMatchKeyword]DecodedMatchKeyword ::= "decmatch"


/* STATIC SEMANTIC – WildcardLengthMatch shall be used when MatchingSymbol is used in fractions of a concatenated string or list (see clause 15.11) and shall not be used in other cases. In this case, the Complement, ListOfTemplates, Range, BitStringMatch, HexStringMatch, OctetStringMatch, CharStringMatch, SubsetMatch and SupersetMatch productions shall not be used. */ 
[bookmark: TExtraMatchingAttributes]ExtraMatchingAttributes ::= StringLength | 
                                 IfPresentKeyword | 
                                 (StringLength IfPresentKeyword) 
[bookmark: TBitStringMatch]BitStringMatch ::= "'" {BinOrMatch} "'" "B" 
[bookmark: TBinOrMatch]BinOrMatch ::= Bin | 
                    AnyValue | 
                    AnyOrOmit 
[bookmark: THexStringMatch]HexStringMatch ::= "'" {HexOrMatch} "'" "H" 
[bookmark: THexOrMatch]HexOrMatch ::= Hex | 
                    AnyValue | 
                    AnyOrOmit 
[bookmark: TOctetStringMatch]OctetStringMatch ::= "'" {OctOrMatch} "'" "O" 
[bookmark: TOctOrMatch]OctOrMatch ::= Oct | 
                    AnyValue | 
                    AnyOrOmit 
[bookmark: TCharStringMatch]CharStringMatch ::= PatternKeyword [CaseInsenModifier] PatternParticle {"&" PatternParticle} 
[bookmark: TPatternParticle]PatternParticle ::= Pattern | ReferencedValue 
[bookmark: TPatternKeyword]PatternKeyword ::= "pattern" 
[bookmark: TPattern]Pattern ::= """ {PatternElement} """ 
[bookmark: TPatternElement]PatternElement ::= (("\" ("?" | "*" | "\" | "[" | "]" | "{" | "}" | 
                               """ | "|" | "(" | ")" | "#" | "+" | "d" | 
                               "w" | "t" | "n" | "r" | "s" | "b" 
                              )) | ("?" | "*" | "\" | "|" | "+" 
                                   ) | ("[" ["^"] [{PatternClassChar ["-"   
                                                                      PatternClassChar]}]   
                                        "]") | 
                         ("{" ["\"] ReferencedValue "}") | ("\" "N" "{"   
                                                           (ReferencedValue | 
                                                           Type) "}") | 
                         (""" """) | 
                         ("(" PatternElement ")") | 
                         ("#" (Num | 
                              ("(" Number "," [Number] ")") | 
                              ("(" "," Number ")") | 
                              ("(" [","] ")") Num ")" 
                             )) 
                        ) | PatternChar 
[bookmark: TPatternChar]PatternChar ::= NonSpecialPatternChar | PatternQuadruple 

/* STATIC SEMANTICS: Characters "?", "*", "\", "[", "]", "{", "}", """, "|", "(", ")", "#", "+", "d", "^", "N" have special semantics – they are metacharacters for the definition of pattern elements – only if they follow the BNF as defined above, if not they are interpreted like normal characters */ 
[bookmark: TNonSpecialPatternChar]NonSpecialPatternChar ::= Char 
[bookmark: TPatternClassChar]PatternClassChar ::= NonSpecialPatternClassChar | 
                          PatternQuadruple | 
                          "\" EscapedPatternClassChar 
[bookmark: TNonSpecialPatternClassChar]NonSpecialPatternClassChar ::= Char 

/* STATIC SEMANTICS: Characters "[", "-", "^", "]", "\", "q", ","have special semantics – they are metacharacters for the definition of pattern class characters – only if they follow the BNF as defined above, if not they are interpreted like normal characters */ 
[bookmark: TEscapedPatternClassChar]EscapedPatternClassChar ::= "[" | "-" | "^" | "]" 
[bookmark: TPatternQuadruple]PatternQuadruple ::= "\" "q" "(" Number "," Number "," Number ","   
                          Number ")" 
[bookmark: TComplement]Complement ::= ComplementKeyword ListOfTemplates 
[bookmark: TComplementKeyword]ComplementKeyword ::= "complement" 
[bookmark: TListOfTemplates]ListOfTemplates ::= "(" TemplateListItem {"," TemplateListItem} ")" 
[bookmark: TTemplateListItem]TemplateListItem ::= TemplateBody | AllElementsFrom 
[bookmark: TAllElementsFrom]AllElementsFrom ::= AllKeyword FromKeyword TemplateBody 
[bookmark: TSubsetMatch]SubsetMatch ::= SubsetKeyword ListOfTemplates 
[bookmark: TSubsetKeyword]SubsetKeyword ::= "subset" 
[bookmark: TSupersetMatch]SupersetMatch ::= SupersetKeyword ListOfTemplates 
[bookmark: TSupersetKeyword]SupersetKeyword ::= "superset" 
[bookmark: TPermutationMatch]PermutationMatch ::= PermutationKeyword ListOfTemplates 

/* STATIC SEMANTICS: Restrictions on the content of TemplateBody within the ListOfTemplates are given in clause B.1.3.3. */ 
[bookmark: TPermutationKeyword]PermutationKeyword ::= "permutation" 
[bookmark: TAnyValue]AnyValue ::= "?" 
[bookmark: TAnyOrOmit]AnyOrOmit ::= "*" 
[bookmark: TWildcardLengthMatch]WildcardLengthMatch ::= LengthKeyword "(" SingleExpression ")" 

/* STATIC SEMANTICS: SingleExpression shall evaluate to type integer */ 
[bookmark: TIfPresentKeyword]IfPresentKeyword ::= "ifpresent" 
[bookmark: TPresentKeyword]PresentKeyword ::= "present" 
[bookmark: TRange]Range ::= "(" Bound ".." Bound ")" 
[bookmark: TBound]Bound ::= (["!"] SingleExpression) | ([Minus] InfinityKeyword) 

/* STATIC SEMANTICS - Bounds shall evaluate to types integer, charstring, universal charstring or float. In case they evaluate to types charstring or universal charstring, the string length shall be 1. infinity as lower bound and –infinity as upper bound are allowed for float types only. */ 
[bookmark: TInfinityKeyword]InfinityKeyword ::= "infinity" 
[bookmark: TActualParAssignment]ActualParAssignment ::= Identifier ":=" TemplateInstance 
[bookmark: TTemplateRefWithParList]/* STATIC SEMANTICS – if a value parameter is used, an in-line template shall evaluate to a value */ TemplateRefWithParList ::= ExtendedIdentifier [ActualParList] 
[bookmark: TTemplateInstance]TemplateInstance ::= [(Type | Signature) Colon] 
TemplateBody 
[bookmark: TDerivedRefWithParList]DerivedTemplateBody ::= ModifiesKeyword BaseTemplateBody AssignmentChar BaseTemplateBody
[bookmark: TActualParList]ActualParList ::= "(" [(ActualPar {"," ActualPar })
                                   {"," ActualParAssignment} |
                                 (ActualParAssignment {"," ActualParAssignment})]
                      ")" 
[bookmark: TActualPar]ActualPar ::= TemplateInstance | Minus 

/* STATIC SEMANTICS - When the corresponding formal parameter is not of template type the TemplateInstance production shall resolve to one or more SingleExpressions */ 
[bookmark: TTemplateOps]TemplateOps ::= MatchOp | ValueofOp 
[bookmark: TMatchOp]MatchOp ::= MatchKeyword "(" Expression "," TemplateInstance ")" 
[bookmark: TMatchKeyword]MatchKeyword ::= "match" 
[bookmark: TValueofOp]ValueofOp ::= ValueofKeyword "(" TemplateInstance")" 
[bookmark: TValueofKeyword]ValueofKeyword ::= "valueof" 


[bookmark: _Toc7508825]A.1.6.9	Miscellaneous productions
[bookmark: TDot]Dot ::= "." 
[bookmark: TMinus]Minus ::= "-" 
[bookmark: TSemiColon]SemiColon ::= ";" 
[bookmark: TColon]Colon ::= ":" 
[bookmark: TUnderscore]Underscore ::= "_" 
[bookmark: TAssignmentChar]AssignmentChar ::= ":=" 
[bookmark: TIndexModifier]IndexModifier ::= "@index" 
[bookmark: TDeterministicModifier]DeterministicModifier ::= "@deterministic" 
[bookmark: TLazyModifier]LazyModifier ::= "@lazy" 
[bookmark: TFuzzyModifier]FuzzyModifier ::= "@fuzzy" 
[bookmark: TCaseInsenModifier]CaseInsenModifier ::= "@nocase"
[bookmark: TDecodedModifier]DecodedModifier ::= "@decoded" 
[bookmark: TDefaultModifier]DefaultModifier ::= "@default"
[bookmark: TControlModifier]ControlModifier ::= "@control" 
AbstractModifier ::= "@abstract" 




G.15	Partially initialized global and local templates
Previous versions of the present document (up to and including V4.11.1) allowed global and local templates without the @abstract modifier to be partially initialized after their initialization has been completed. This feature is deprecated and may be fully removed in a future edition of the present document.

